
Abstract 
- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - 

このテクニカルノートでは、4th DimensionのDOMコマンドを使用して、XMLノード

のフルパスを構築する方法を示します。またサンプルデータベースには実装済みのコー

ドが含まれています。 

 

Overview 
- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - 

XMLは、システム間のデータ交換や、プログラム設定およびその他データの格納フォー

マットなど、データフォーマットのスキームとして、たいへん身近なものとなりました。

XMLを使用する際の難関のひとつは、情報を階層的に格納し体系化するという、XML

の最大の機能の中にあります。例として、以下の短いXMLを見てみましょう: 
 
<?xml version="1.0" encoding="UTF-8" standalone="no" ?> 
<contacts> 
 <contact> 
  <name>Dan West</name> 
  <business> 
   <phone> 
    <area_code>04</area_code> 
    <number>0035-9110</number> 
   </phone> 
  </business> 
  <home> 
   <phone> 
    <area_code>02</area_code> 
    <number>6493-3250</number> 
   </phone> 
  </home> 
 </contact> 
</contacts> 
 
この例では、以下の要素が2回づつ登場しています: 

 
<phone> 
<area_code> 
<number> 
 
これらの要素は<business>と<home>要素の子要素として登場します。XMLドキュメ

ントを処理する際、要素名が繰り返し現れるのは普通のことです。このような場合、ノ

ードへのフルパスを取得することで、それらを区別することができるようになります。

例えば<area_code>への2つのインスタンスへのパスは、以下のようになります: 

 
/contacts/contact/business/phone/area_code/ 
/contacts/contact/home/phone/area_code/ 
 
このテクニカルノートでは、ノードへのフルパスを取得する方法を説明しています。ま

たコードが実装されたサンプルデータベースを含んでいます。 



 

Review: Native DOM Commands 
- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - 

4th Dimensionにはすでに、パスを使用するDOM (Document Object Model) コマ

ンドが含まれています。例えば上に示した2つの<area_code>要素にアクセスするこ

とは簡単で、以下のように行います: 

 
C_STRING($16;$businessPhone_xmlref) 
C_STRING($16;$homePhone_xmlref) 
 
$businessPhone_xmlref:=DOM Find XML element($root_xmlref; 
     "/contacts/contact/business/phone/area_code/") 
 
$homePhone_xmlref:=DOM Find XML element($root_xmlref; 
     "/contacts/contact/home/phone/area_code/") 
 

DOM Find XML element 関数は簡単・有効ですが、事前にフルパスが分かっていない

と使用することができません。このコマンドはほとんどのXML処理で使用することがで

きますが、すべてではありません。例えば: 

 

■ XMLは一般に木構造を上から下にたどっていく方法で検査されます。このテクニッ

クは既存のテクニカルノートやサンプルデータベース、およびThe 4D Web Services 

Companionで使用されています。この方法を使用した場合、フルパスを最初に知ること

はできません。代わりに、コードは順番にそれぞれのノードを処理していきます。DOM 

Find XML element関数は、木構造のトップから下る方向で処理を行います。対して一

般の処理は、ノードから登る方向で木構造をたどります。 

 

■ DOM Find XML element関数のパスシンタックスは、すべてのマッチパターンを表

現することはできません。例えば、この関数は/phone/area_code/で終了するパス

といった条件を使用することができません。ドキュメントルートやフルパスがどうであ

るかにかかわらず、ノードや木構造の連続した一部分に基づいて処理を行うと便利であ

ったり、それが必要な場合があります。 

 

幸いなことに、他のコマンドを使用して、簡単にフルパスを生成する関数を記述するこ

とができます。 

 

Note XMLパス、要素名、および属性名は大文字小文字を区別します。このような場合の文字列

比較については4D Technical Note 05-41, Case-Sensitive Operations in 4th Dimensionを参

照してください。 

 

 



The Routines 
- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - 

DOM_GetFullPathルーチンのコードとエラーハンドラを以下に示します。 

 

DOM_GetFullPath 
以下のコードは、カレントノードのフルパスを返します。無効なノードやヌルを渡すと、

空の文字が返されます。 
 
C_TEXT($0;$fullPath_text) 
C_STRING(16;$1;$noderef)  ` <-- The calling routine must pass in this parameter. 
 
$noderef:=$1 
$fullPath_text:="" 
 
` Store existing error/error handling state. 
If (Undefined(Error)) 
  Error:=0 
End if 
 
C_LONGINT($previousValueOfErrorVariable_l) 
$previousValueOfErrorVariable_l:=Error 
 
C_STRING(31;$previousErrorMethodName_s) 
$previousErrorMethodName_s:=Method called on error 
 
ON ERR CALL("DOM_ErrorTrappingRoutine") 
 
 
C_TEXT($name_text) 
$name_text:="" 
DOM GET XML ELEMENT NAME($noderef;$name_text)  ` Default to starting element name. 
 
While (OK=1) 
  Case of 
    ¥ ($name_text="#document")  ` The #document node is a synthetic node above the tree. 
      OK:=0  ` Get out of routine. 
 
    ¥ ($name_text="")  ` Bad reference. 
      OK:=0  ` Get out of routine. 
 
    Else  ` Valid name. 
      $fullPath_text:="/"+$name_text+$fullPath_text 
      OK:=1  ` Continue processing. 
      $noderef:=DOM Get parent XML element($noderef;$name_text)  ` Try to get a parent. 
 
  End case 
End while 
 
  ` Restore previous error/error handling state. 
Error:=$previousValueOfErrorVariable_l  `Restore original error value. 
ON ERR CALL($previousErrorMethodName_s)  `Restore original error handler. 
 
If ($fullPath_text#"")  ` A path was built. 
  $fullPath_text:=$fullPath_text+"/"  ` Add final slash. 
End if 
 
  ` Return result. 
$0:=$fullPath_text 

 

 

 



DOM_ErrorTrappingRoutine 
DOM_ErrorTrappingRoutineはDOM_GetFullPathによってインストールされ、無効なノ

ードを参照することによって生成されるエラーをトラップします。エラーハンドラは以

下のようなシンプルなコードです: 

 
DOM_Error:=Error 
 

The Sample Database 
- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - 

サンプルデータベースには、先に説明したコードと、単純なテストルーチン

Test_GetFullPathが含まれています。テストコードは有効なノードや無効なノードを含

むさまざまな条件でDOM_GetFullPathを実行するようデザインされています。テスト

ごとに、コードは結果を以下のようなダイアログに表示します: 

 

 

 

Summary 
- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - 

4th DimensionのDOM XMLコマンドには、特定のXMLノードへのフルパスを返す関

数が含まれていません。しかしながら、この関数を、DOM Get parent XML element

コマンドを使用して簡単に実装することができます。サンプルデータベースには

OM_GetFullPathという名前の、フルパスを返すメソッドが含まれています。 


